// Vulnerability label: clean
pragma solidity ^0.4.23;

/**
 * @title SafeMath
 * @dev Math operations with safety checks that throw on error
 */
library SafeMath {

  /**
  * @dev Multiplies two numbers, throws on overflow.
  */
  function mul(uint256 a, uint256 b) internal pure returns (uint256 c) {
    if (a == 0) {
      return 0;
    }
    c = a * b;
    assert(c / a == b);
    return c;
  }

  /**
  * @dev Integer division of two numbers, truncating the quotient.
  */
  function div(uint256 a, uint256 b) internal pure returns (uint256) {
    // assert(b > 0); // Solidity automatically throws when dividing by 0
    // uint256 c = a / b;
    // assert(a == b * c + a % b); // There is no case in which this doesn't hold
    return a / b;
  }

  /**
  * @dev Subtracts two numbers, throws on overflow (i.e. if subtrahend is greater than minuend).
  */
  function sub(uint256 a, uint256 b) internal pure returns (uint256) {
    assert(b <= a);
    return a - b;
  }

  /**
  * @dev Adds two numbers, throws on overflow.
  */
  function add(uint256 a, uint256 b) internal pure returns (uint256 c) {
    c = a + b;
    assert(c >= a);
    return c;
  }
}

/**
 * @title ERC20Basic
 * @dev Simpler version of ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/179
 */
contract ERC20Basic {
  function totalSupply() public view returns (uint256);
  function balanceOf(address who) public view returns (uint256);
  function transfer(address to, uint256 value) public returns (bool);
  event Transfer(address indexed from, address indexed to, uint256 value);
}

/**
 * @title ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/20
 */

contract ERC20 is ERC20Basic {
  function allowance(address owner, address spender)
    public view returns (uint256);

  function transferFrom(address from, address to, uint256 value)
    public returns (bool);

  function approve(address spender, uint256 value) public returns (bool);
  event Approval(
    address indexed owner,
    address indexed spender,
    uint256 value
  );
}

/**
 * @title Basic token
 * @dev Basic version of StandardToken, with no allowances.
 */
contract BasicToken is ERC20Basic {
  using SafeMath for uint256;

  mapping(address => uint256) balances;

  uint256 totalSupply_;

  /**
  * @dev total number of tokens in existence
  */
  function totalSupply() public view returns (uint256) {
    return totalSupply_;
  }

  /**
  * @dev transfer token for a specified address
  * @param _to The address to transfer to.
  * @param _value The amount to be transferred.
  */
  function transfer(address _to, uint256 _value) public returns (bool) {
    require(_to != address(0));
    require(_value <= balances[msg.sender]);

    balances[msg.sender] = balances[msg.sender].sub(_value);
    balances[_to] = balances[_to].add(_value);
    emit Transfer(msg.sender, _to, _value);
    return true;
  }

  /**
  * @dev Gets the balance of the specified address.
  * @param _owner The address to query the the balance of.
  * @return An uint256 representing the amount owned by the passed address.
  */
  function balanceOf(address _owner) public view returns (uint256) {
    return balances[_owner];
  }

}

/**
 * @title Standard ERC20 token
 *
 * @dev Implementation of the basic standard token.
 * @dev https://github.com/ethereum/EIPs/issues/20
 * @dev Based on code by FirstBlood: https://github.com/Firstbloodio/token/blob/master/smart_contract/FirstBloodToken.sol
 */
contract StandardToken is ERC20, BasicToken {

  mapping (address => mapping (address => uint256)) internal allowed;


  /**
   * @dev Transfer tokens from one address to another
   * @param _from address The address which you want to send tokens from
   * @param _to address The address which you want to transfer to
   * @param _value uint256 the amount of tokens to be transferred
   */
  function transferFrom(
    address _from,
    address _to,
    uint256 _value
  )
    public
    returns (bool)
  {
    require(_to != address(0));
    require(_value <= balances[_from]);
    require(_value <= allowed[_from][msg.sender]);

    balances[_from] = balances[_from].sub(_value);
    balances[_to] = balances[_to].add(_value);
    allowed[_from][msg.sender] = allowed[_from][msg.sender].sub(_value);
    emit Transfer(_from, _to, _value);
    return true;
  }

  /**
   * @dev Approve the passed address to spend the specified amount of tokens on behalf of msg.sender.
   *
   * Beware that changing an allowance with this method brings the risk that someone may use both the old
   * and the new allowance by unfortunate transaction ordering. One possible solution to mitigate this
   * race condition is to first reduce the spender's allowance to 0 and set the desired value afterwards:
   * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
   * @param _spender The address which will spend the funds.
   * @param _value The amount of tokens to be spent.
   */
  function approve(address _spender, uint256 _value) public returns (bool) {
    allowed[msg.sender][_spender] = _value;
    emit Approval(msg.sender, _spender, _value);
    return true;
  }

  /**
   * @dev Function to check the amount of tokens that an owner allowed to a spender.
   * @param _owner address The address which owns the funds.
   * @param _spender address The address which will spend the funds.
   * @return A uint256 specifying the amount of tokens still available for the spender.
   */
  function allowance(
    address _owner,
    address _spender
   )
    public
    view
    returns (uint256)
  {
    return allowed[_owner][_spender];
  }

  /**
   * @dev Increase the amount of tokens that an owner allowed to a spender.
   *
   * approve should be called when allowed[_spender] == 0. To increment
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   * @param _spender The address which will spend the funds.
   * @param _addedValue The amount of tokens to increase the allowance by.
   */
  function increaseApproval(
    address _spender,
    uint _addedValue
  )
    public
    returns (bool)
  {
    allowed[msg.sender][_spender] = (
      allowed[msg.sender][_spender].add(_addedValue));
    emit Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

  /**
   * @dev Decrease the amount of tokens that an owner allowed to a spender.
   *
   * approve should be called when allowed[_spender] == 0. To decrement
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   * @param _spender The address which will spend the funds.
   * @param _subtractedValue The amount of tokens to decrease the allowance by.
   */
  function decreaseApproval(
    address _spender,
    uint _subtractedValue
  )
    public
    returns (bool)
  {
    uint oldValue = allowed[msg.sender][_spender];
    if (_subtractedValue > oldValue) {
      allowed[msg.sender][_spender] = 0;
    } else {
      allowed[msg.sender][_spender] = oldValue.sub(_subtractedValue);
    }
    emit Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

}

/**
 * @title Ownable
 * @dev The Ownable contract has an owner address, and provides basic authorization control
 * functions, this simplifies the implementation of "user permissions".
 */
contract Ownable {
  address public owner;


  event OwnershipRenounced(address indexed previousOwner);
  event OwnershipTransferred(
    address indexed previousOwner,
    address indexed newOwner
  );


  /**
   * @dev The Ownable constructor sets the original `owner` of the contract to the sender
   * account.
   */
  constructor() public {
    owner = msg.sender;
  }

  /**
   * @dev Throws if called by any account other than the owner.
   */
  modifier onlyOwner() {
    require(msg.sender == owner);
    _;
  }

  /**
   * @dev Allows the current owner to transfer control of the contract to a newOwner.
   * @param newOwner The address to transfer ownership to.
   */
  function transferOwnership(address newOwner) public onlyOwner {
    require(newOwner != address(0));
    emit OwnershipTransferred(owner, newOwner);
    owner = newOwner;
  }

  /**
   * @dev Allows the current owner to relinquish control of the contract.
   */
  function renounceOwnership() public onlyOwner {
    emit OwnershipRenounced(owner);
    owner = address(0);
  }
}

contract MNC is StandardToken, Ownable {
  string public constant name = "Moneynet Coin";
  string public constant symbol = "MNC";
  uint32 public constant decimals = 18;

  // Tokens amount
  uint256 public saleTokens = uint256(6e9).mul(1 ether);
  uint256 public ecosystemTokens = uint256(204e8).mul(1 ether);
  uint256 public teamTokens = uint256(4e9).mul(1 ether);
  uint256 public investorsTokens = uint256(24e8).mul(1 ether);
  uint256 public advisorsTokens = uint256(2e9).mul(1 ether);
  uint256 public bonusTokens = uint256(16e8).mul(1 ether);
  uint256 public reserveTokens = uint256(36e8).mul(1 ether);

  // sale contract address
  address public saleContract;

  // locked tokens data
  mapping(address => uint256) public lockedTokens_3;
  mapping(address => uint256) public lockedTokens_6;
  mapping(address => uint256) public lockedTokens_12;
  uint256 lockTime = now;

  constructor(address _newOwner) public {
    require(_newOwner != address(0));
    uint256 tokens = ecosystemTokens;
    owner = _newOwner;
    balances[owner] = balances[owner].add(tokens);
    totalSupply_ = totalSupply_.add(tokens);
    emit Transfer(address(0), owner, tokens);
  }


  /**
   * @dev Sending tokens for sale to the sale contract address.
   * @param _contract Sale contract address.
   */
  function activateSaleContract(address _contract) public onlyOwner returns (bool) {
    require(_contract != address(0));
    require(saleTokens > 0);
    uint256 tokens = saleTokens;
    saleTokens = 0;
    saleContract = _contract;
    totalSupply_ = totalSupply_.add(tokens);
    balances[_contract] = balances[_contract].add(tokens);
    emit Transfer(address(0), _contract, tokens);
    return true;
  }

  /**
   * @dev Sending reserve tokens to the sale contract address.
   */
  function sendReserveTokens() public onlyOwner returns (bool) {
    require(saleContract != address(0));
    require(reserveTokens > 0);
    uint256 tokens = reserveTokens;
    reserveTokens = 0;
    totalSupply_ = totalSupply_.add(tokens);
    balances[saleContract] = balances[saleContract].add(tokens);
    emit Transfer(address(0), saleContract, tokens);
    return true;
  }

  /**
   * @dev  Team tokens accrual.
   * @param _address Recepient address.
   * @param _amount Accrued amount.
   */
  function accrueTeamTokens(address _address, uint256 _amount) public onlyOwner returns (bool) {
    require(_amount > 0);
    require(_amount <= teamTokens);
    require(_address != address(0));
    teamTokens = teamTokens.sub(_amount);
    lockedTokens_12[_address] = lockedTokens_12[_address].add(_amount);
    return true;
  }

  /**
   * @dev  Investors tokens accrual.
   * @param _address Recepient address.
   * @param _amount Accrued amount.
   */
  function accrueInvestorsTokens(address _address, uint256 _amount) public onlyOwner returns (bool) {
    require(_amount > 0);
    require(_amount <= investorsTokens);
    require(_address != address(0));
    investorsTokens = investorsTokens.sub(_amount);
    lockedTokens_6[_address] = lockedTokens_6[_address].add(_amount);
    return true;
  }

  /**
   * @dev  Advisors tokens accrual.
   * @param _address Recepient address.
   * @param _amount Accrued amount.
   */
  function accrueAdvisorsTokens(address _address, uint256 _amount) public onlyOwner returns (bool) {
    require(_amount > 0);
    require(_amount <= advisorsTokens);
    require(_address != address(0));
    advisorsTokens = advisorsTokens.sub(_amount);
    lockedTokens_6[_address] = lockedTokens_6[_address].add(_amount);
    return true;
  }

  /**
   * @dev  Bonus tokens accrual.
   * @param _address Recepient address.
   * @param _amount Accrued amount.
   */
  function accrueBonusTokens(address _address, uint256 _amount) public onlyOwner returns (bool) {
    require(_amount > 0);
    require(_amount <= bonusTokens);
    require(_address != address(0));
    bonusTokens = bonusTokens.sub(_amount);
    lockedTokens_3[_address] = lockedTokens_3[_address].add(_amount);
    return true;
  }

  function releaseTokens() public returns (bool) {
    uint256 tokens = 0;
    if (lockedTokens_3[msg.sender] > 0 && now.sub(lockTime) > 91 days) {
      tokens = tokens.add(lockedTokens_3[msg.sender]);
      lockedTokens_3[msg.sender] = 0;
    }
    if (lockedTokens_6[msg.sender] > 0 && now.sub(lockTime) > 182 days) {
      tokens = tokens.add(lockedTokens_6[msg.sender]);
      lockedTokens_6[msg.sender] = 0;
    }
    if (lockedTokens_12[msg.sender] > 0 && now.sub(lockTime) > 365 days) {
      tokens = tokens.add(lockedTokens_12[msg.sender]);
      lockedTokens_12[msg.sender] = 0;
    }
    require (tokens > 0);
    totalSupply_ = totalSupply_.add(tokens);
    balances[msg.sender] = balances[msg.sender].add(tokens);
    emit Transfer(address(0), msg.sender, tokens);
  }
}